

NATIONAL RADIO ASTRONOMY OBSERVATORY
CHARLOTTESVILLE, VIRGINIA

ELECTRONICS DIVISION INTERNAL REPORT No. 171

MICROWAVE MEASUREMENT PROGRAM FOR
THE H.P. 9830A CALCULATOR

JOHN GRANLUND

SEPTEMBER 1976

NUMBER OF COPIES: 150

This program point-plots on a Smith chart and prints the reflection coefficients of a mixer diode that it establishes in two quite distinct modes:

Mode (1). The microwave incremental impedance of the diode is modeled by the circuit of Figure 1, in which the shunt conductance $G(I)$ and capacitance $C(V)$ are dependent on the diode DC bias current and voltage, as shown in the figure. In this mode, the diode reflection coefficients are calculated from the circuit model at the bias values that have been entered into memory.

Mode (2). Slotted line measurements of the diode are made using the equipment configuration suggested by the block diagram of Figure 2. The transducer joining the input slotted line to the output connecting lines can be a quite arbitrary two-port -- active or lossy, but linear and reciprocal. Four complex parameters are required to describe the transducer at one frequency. These are evaluated first from slotted line calibration measurements obtained with the connecting lines and diode replaced by each of three known and distinct calibrating loads attached at the transducer output reference plane. The fourth condition for determining the transducer parameters comes from assuming the transducer to be reciprocal. An alternate, simplified calibration procedure is also incorporated into the program that requires only one calibrating load. The transducer is then modeled as an attenuator -- or bilateral amplifier -- matched to the input and output lines, in tandem with an unknown length of transmission line. This simplified modeling forces two of the parameters describing the transducer to vanish.

With calibration completed and the equipment configuration of Figure 2 restored, the program next calculates the parameters that describe the composite two-port connecting the slotted line to the diode. In this second mode, diode reflection coefficients are finally calculated using these composite parameters and the slotted line measurement points -- voltage standing wave ratio and probe position -- that have been entered into memory.

In both modes the diode reflection coefficients calculated are referred to 50 ohms: They would result at the diode if it were driven from a 50 ohm line.

The program has been assembled to assist in reducing slotted line measurements of a microwave diode and its mount, to compare these results with the theory associated with the diode model, and to ease the selection of both diode and mount parameters that yield close agreement between the measurements and the theory. To this end, the program is arranged, as far as is practicable, to retain in memory those numbers entered by the operator. The program and the essential input data are stored in separate tape cassette files, so that the outputs obtained the last time the program was used can be duplicated without re-entering data merely by recalling both the program and the data from the cassette files to the calculator memory. If several microwave measurement jobs are simultaneously in progress, or if more data storage is needed for one job, it may be convenient to assign several input data files in the tape cassette. In dealing with each data file, a number of options are also provided for selecting the desired output and for changing only selected portions of the input data: Connecting line parameters can be changed, for example, without re-entering slotted line and bias measurements.

Although the purpose of the program is very specific, it should be remembered that calculations are made in two modes that are essentially independent of each other and can therefore be used separately. In particular, the reduction of slotted line measurements in the second mode does not require a diode load; an old shoe or any linear microwave load could be measured instead, through a transducer whose properties are determined by the calibration.

1. Using the Program

The point plot and full printout obtained from one row of data are shown in Figures 3 and 4, respectively; the program itself is attached at the end of this memo.

Storage space is available for eight "rows" of data, and these rows are kept filled with current, out-of-date, or trumped-up data, at the option of the operator. Each row corresponds naturally to a different frequency, but it is the operator's option to fill several or all eight rows with data taken at the same frequency. In Figure 4, the data of row number 8 is all printed between the headings "INPUT OF ROW NO. 8" and "DIODE MODEL OF ROW NO. 8". With some exceptions, the data of the other rows are separately entered and recalled. The exceptions -- data common to all eight rows -- are as follows:

- (1) The six diode parameters listed under "DIODE MODEL".
- (2) The characteristic impedances and delays listed under "CONNECTING LINES".
- (3) The calibrating load descriptions listed under "GAMMA" and "DELAY", but not those under "VSWR" and "X", of "CALIBRATIONS".

If one of the above values is changed in dealing with row 8, the changed value will be found to apply to each of the other rows. It is, however, possible to use a single-load calibration for some of the rows and a three-load calibration for the others without re-entering calibration data whenever the row is changed; the "single load" must then be the same as the first of the "three loads".

For each row, storage space is available for fifteen points, each represented by the four values I, V, VSWR, and X tabulated under "MEASUREMENTS" in Figure 4. But it is not necessary to enter data for all fifteen points: Data for only four points have been entered in row number 8.

1.1 Plot Details

Once the data for a row have been entered, one or both of the point plots of Figure 3 can be called out. The points marked "0" and connected by straight lines are calculated in Mode (1) from the diode model; the disconnected points marked "X" are calculated from the slotted line measurements in Mode (2). A printout option is provided for producing the printout under "DIODE MODEL OF ROW NO. 8" with the diode model points and the printout under "MEASUREMENTS OF ROW NO. 8" with the slotted line points.

The radius of the outer circle of Smith Chart Form 82-BSPR (9-66) printed by the Kay Electric Company, Pine Brook, N.J. is about 1.14927 units, and the plot area used by the program is the square tangent to this outer circle, set in program steps 50 and 60. This allows calculated reflection coefficients at least as large as 1.14927 in magnitude to be faithfully plotted. It is convenient to draw the tangent square on a Kay Smith Chart Form, Xerox the modified form, and use the copies for preliminary plotting. Upper right and lower left graph limits are the corresponding corners of the tangent square. For a final plot, more accuracy can be obtained by aligning one edge of the square with the corresponding plotter axis. For example, after the program has once been run through step 60, the executed keyboard command "PLOT -1.14927, 1.14927, 1" will locate the pen over the upper left corner. This command and

the "LOWER LEFT" button on the plotter allow the left edge of the tangent square to be aligned with the plotter's y-axis.

The program will run with the plotter turned off, so turning either the plotter's "LINE" or "CHART HOLD" buttons off is an easy way to avoid an unwanted plot. But if the printer is turned off, program execution halts at the first print statement. For this reason, a printout option is incorporated into the program to allow the operator to minimize -- or eliminate altogether -- the printout associated with his preliminary plots without having to deal with a halted program.

1.2 Options

Two sorts of numerical inputs are called for by the program: the data -- diode parameters, VSWR's, delays, etc., which tend to be multidigit numbers requiring considerable care on the part of the operator to avoid keyboard errors -- and the option numbers -- one digit integers used to select the program steps to be executed. The operator can expect to remember the option numbers, eliminating the need for look-up, when he has become familiar with the program. In the familiarization process it is helpful to keep a copy of the program steps through 1160 nearby; all of the options are selected in this part of the program, and remarks have been included to assist the operator in selecting options. The program steps containing all of the options and the more pertinent remarks are listed in Figure 5.

There are eight options. First, the data row -- to be used as is or to be modified -- is selected for producing output.

Second, the amount of printout is specified. P=0 produces no printout, and P=2 produces the full printout of Figure 4; Mode (1) printout will appear only with a Mode (1) plot, and Mode (2) printout will appear only with a Mode (2) plot. P=1 suppresses both Mode (1) and Mode (2) printout and in addition prints only the data that is subsequently changed. As discussed next in connection with options 3 through 7, the data is subdivided into five blocks, and an entire block is printed if any of its data are changed.

Third, a choice is made as to whether the connecting line parameters are to be changed. Here, as in the next two options, a three-way choice is offered. K=0 moves the program out of the data-changing mode and into the printout mode; no more data changes can then be made without re-running the program. K=1 moves

the program on to the next data block without offering changes in the present block. K=2 calls for changes in the present block. All four connecting line parameters -- two characteristic impedances and two one-way delays -- must be entered. As noted above, these changes will affect all eight rows of data.

Fourth, the six diode model parameters are offered for change. The same choices as above -- K=0, 1, 2 -- are available. Changes here also affect all eight rows of data.

Fifth, the "frequency" block is offered for change, again with choices K=0,1 and 2. But because the row number has already been specified, and the number of measured points is counted by the program, only frequency and slotted line guide wavelength need be entered. Guide wavelength must be expressed in the same units as slotted line probe position.

Sixth, the "calibrations" block is offered for change in a four-way choice. M=0 moves the program out of the data-changing mode and M=1 moves the program on to the next data block, both without offering calibration changes. M=2 offers changes in the three VSWR's and probe positions measured with the three calibrating loads. M=3 offers, in addition, changes in the calibrating loads themselves; these latter changes will affect all eight rows of data. The calibrating loads are specified by a real reflection coefficient, GAMMA, and the one-way delay of the 50 ohm line between the transducer output reference plane and the point in this line where the reflection coefficient is GAMMA. This method of load specification is well suited to the use of shorts and matched loads for calibration, but any three distinct loads -- active or passive and with positive or negative delay -- can be used. When changing data in the "calibrations" block, a single-load calibration is achieved by describing that single load as the first load and inputting VSWR=999 when the second load description is called for. The printout will then show only one calibrating load.

Seventh, the "measurements" block is offered for change in a four-way choice that is slightly different from the one just described. M=0 exits the data-changing mode. M=1 offers the diode bias currents and voltages for change. M=2 offers the corresponding slotted line VSWR's and probe positions for change. M=3 offers both DC and RF quantities for change. Although storage space is available for fifteen points in each row, data entry can also be limited to any

lesser number of points by entering $I < 0$ when $M=1$ or 3 and/or $VSWR=999$ when $M=2$ or 3 , following the last valid data point entry. Printout, as selected in the second option, follows.

Eighth, the output is selected in a four-way choice. Other than the above printout, $M=0$ produces no output. $M=1$ produces a plot of Mode (1) points, together with the printout selected in the second option, if any. Similarly, $M=2$ produces a Mode (2) point plot, and $M=3$ produces both plots. In all cases, the program line counter is reset by the END statement after the selected output has been produced.

1.3 Miscellaneous Notes

In recalling program and data from the tape cassette files to the calculator memory, it is necessary to LOAD the program first. Before switching the calculator off or changing data files, remember to STORE in its cassette file the DATA that has been collected in the calculator memory. In changing data files, merely LOAD the new DATA file from tape; the old data in the calculator memory will be written over by the new data, and the program will remain intact.

As of this writing, 3234 words of storage are required for the program and 2292 words are required for each data file. File sizes of 3500 and 2500 words, respectively, have been found adequate.

Because of the length of the program, several seconds is required for initialization. This is performed each time program execution is initiated by the keyboard commands RUN, EXECUTE. But initialization is only required for the first execution after the program has been recalled from the cassette file. Since the program line counter is reset at the end of each execution, initialization for subsequent executions can be avoided by using the keyboard commands CONT, EXECUTE to initiate them.

2. Computational Basis

○

This section exposes the fundamental principles upon which the computations are based and describes some of their salient features. Completeness has been avoided in favor of brevity.

2.1 Mode (1)

The intrinsic capacitance of a semiconductor diode varies with bias voltage according to

$$C(V) = C(1 - V/V_2)^{-N}$$

when $V \ll V_2$. When V becomes positive, approaching -- and exceeding -- V_2 , more RF current flows in the intrinsic conductance than in the intrinsic capacitance, making the capacitance difficult to measure. The expression used by the program,

$$C(V) = \begin{cases} C(1 - V/V_2)^{-N} , & V/V_2 < 1/2 \\ C[2 \cdot 2^N - (V/V_2)^{-N}] , & V/V_2 \geq 1/2 \end{cases} \quad (1)$$

agrees with the first expression when $V < 1/2 V_2$ and continues it in an odd-symmetric fashion above $V = 1/2 V_2$ to avoid the singularity of the first expression. This allows the impedance of the intrinsic diode to be controlled by the conductance, as is suggested by measurement, rather than being shorted at $V = V_2$ by the infinite capacitance of the first expression.

According to Figure 1, the impedance at the diode reference plane is

$$Z = R + j 2\pi f L + 1/(G(I) + j 2\pi f C(V)) \quad (2)$$

and the reflection coefficient, referred to 50 ohms, is

$$\rho = \frac{Z - 50}{Z + 50} \quad (3)$$

These are the expressions implemented by the program.

2.2 Mode (2)

In a voltage-and-current description of two-terminal-pair networks, the so-called general circuit parameters A, B, C, D have the following useful properties:

(1) If the impedance Z_2 is attached to terminal-pair 2, the impedance seen looking in at terminal-pair 1 is

$$Z_1 = \frac{A Z_2 + B}{C Z_2 + D}$$

Or alternately, the impedance Z_1 measured at terminal-pair 1 must be the result of attaching the impedance

$$Z_2 = \frac{B - D Z_1}{C Z_1 - A}$$

to terminal-pair 2. Both of these relations are Möbius transformations of one impedance into the other.

(2) The A, B, C, D matrix of two two-terminal-pair networks in tandem is the (matrix) product of their individual A, B, C, D matrices.

(3) The network is reciprocal if and only if the determinant

$$\left| \begin{bmatrix} A & B \\ C & D \end{bmatrix} \right| = A D - B C = 1$$

In a wave description of two-ports, the transformation - or T-matrix

$$[T] = \begin{bmatrix} T_{11} & T_{12} \\ T_{21} & T_{22} \end{bmatrix}$$

and its elements, the T-parameters, also possess the above properties, with the exception that the impedances of (1) are replaced by reflection coefficients. For these reasons the T-parameters, with a simplifying modification, are used by the program to describe the transducer of Figure 2 and its various tandem combinations with the connecting lines.

Specifically, the T-parameters -- before the modification used by the program -- are defined implicitly by the way in which they interrelate the complex amplitudes of the incident and reflected waves at the output reference plane to those at the input reference plane, as shown in Figure 6:

$$\begin{bmatrix} \alpha \\ \beta \end{bmatrix} = [T] \times \begin{bmatrix} a \\ b \end{bmatrix} \quad (4)$$

The reflection coefficient at the output reference plane is then related to the reflection coefficient $r = b/a$ at the input reference plane according to

$$\rho = \frac{\beta}{\alpha} = \frac{T_{21} a + T_{22} b}{T_{11} a + T_{12} b} = \frac{T_{21} + T_{22} r}{T_{11} + T_{12} r} \quad (5)$$

A similar expression is used to calculate the diode reflection coefficients from the slotted line measurements and the T-parameters of the composite two-port between the slotted line and the diode reference plane.

From the definition provided by equation (4) and Figure 6, the T-matrix of the tandem combination of transducers of Figure 7 is evidently

$$[T] = [T_3] \times [T_2] \times [T_1] \quad (6)$$

with the obvious extension when more transducers are combined. The program uses similar matrix multiplications to calculate, from the T-parameters of the component two-ports, the T-parameters of the composite two-port connecting the slotted line to the diode.

The scattering matrix of a transducer specified by T-parameters is

$$[S] = \frac{1}{T_{22}} \cdot \begin{bmatrix} -T_{21} & 1 \\ |[T]| & T_{12} \end{bmatrix}$$

and the transducer is reciprocal if and only if the scattering matrix is symmetrical and therefore if and only if the T-matrix has a unit determinant:

$$|[T]| = T_{11} T_{22} - T_{12} T_{21} = 1 \quad (7)$$

This is used as one of the four conditions for evaluating the T-parameters of the transducer of Figure 2. The other three conditions, which are applications of equation (5), read

$$\rho_k = \frac{T_{21} + T_{22} r_k}{T_{11} + T_{12} r_k}, \quad k = 1, 2, 3 \quad (8)$$

in which the three ρ 's are the known reflection coefficients of the calibrating loads at the transducer output reference plane, and the three r 's are the corresponding reflection coefficients at the input reference plane obtained from the slotted line measurements.

A solution of the four equations (7) and (8) reads

$$T_{22} = \frac{1}{D} [(\rho_3 - \rho_1)(r_2 \rho_2 - r_1 \rho_1) - (\rho_2 - \rho_1)(r_3 \rho_3 - r_1 \rho_1)]$$

$$T_{11} = \frac{1}{D} [(r_3 - r_1)(r_2 \rho_2 - r_1 \rho_1) - (r_2 - r_1)(r_3 \rho_3 - r_1 \rho_1)]$$

$$T_{12} = \frac{1}{D} [(r_2 - r_1)(\rho_3 - \rho_1) - (r_3 - r_1)(\rho_2 - \rho_1)]$$

$$T_{21} = T_{11} \rho_1 + r_1 (T_{12} \rho_1 - T_{22})$$

in which

$$D = \sqrt{(r_3 - r_2)(r_3 - r_1)(r_2 - r_1)(\rho_3 - \rho_2)(\rho_3 - \rho_1)(\rho_2 - \rho_1)}$$

The sign ambiguity introduced by the square root should raise doubts as to the validity of the solution! But a more careful inspection shows all T-parameters to be proportional without ambiguity to D^{-1} . The net result is therefore a single ambiguity in the sign of the T-matrix. Doubts about the solution validity are then removed by reference to equation (5), which shows no change in calculated output reflection coefficient when the signs of all T-parameters are changed. But why should there be any ambiguity? Figure 6 and equation (4) show that the effect of inserting an ideal -1:1 transformer in the transducer output line is to change the sign of all four T-parameters. Clearly this insertion can have no effect on either the output or input reflection coefficients. But no set of measurements made entirely on the transducer input line without input-output phase comparison can detect the presence of the transformer.

The investigation of the ambiguity has only started. Its conclusion produces the simplification yielding the modified T-parameters contained in the matrix

$$[t] = \begin{bmatrix} t_{11} & t_{12} \\ t_{21} & t_{22} \end{bmatrix} \propto [T]$$

and used by the program. The evidence is as follows:

(1) The T-parameters of the composite two-port will be used only as in equation (5) to calculate various output reflection coefficients.

(2) The result of such a calculation is not affected if the T-parameters are all multiplied by the same (finite, non-zero) complex number.

(3) Since, according to equation (6), the T-matrix of the composite two-port is the product of the T-matrices of the component two-ports, each component T-matrix may also be multiplied by an arbitrary complex number without affecting the calculated results.

This discussion will be completed by determining the T-matrices of the component two-ports suggested by Figure 2, selecting multipliers that yield the simplest t-matrices, and exhibiting the modified T-parameters -- the t's -- used by the program. For the transducer itself the obvious choice is $[t] = D [T]$, with the t-parameters

$$\left. \begin{aligned} t_{22} &= (\rho_3 - \rho_1)(r_2 \rho_2 - r_1 \rho_1) - (\rho_2 - \rho_1)(r_3 \rho_3 - r_1 \rho_1) \\ t_{11} &= (r_3 - r_1)(r_2 \rho_2 - r_1 \rho_1) - (r_2 - r_1)(r_3 \rho_3 - r_1 \rho_1) \\ t_{12} &= (r_2 - r_1)(\rho_3 - \rho_1) - (r_3 - r_1)(\rho_2 - \rho_1) \\ t_{21} &= t_{11} \rho_1 + r_1(t_{12} \rho_1 - t_{22}) \end{aligned} \right\} \quad (9)$$

Of the component two-ports of Figure 2, the one that tends to be pushed aside and neglected, perhaps because it seems to be transparent, is the zero-electrical-length transition from one characteristic impedance to another.

Figure 2 shows three such transitions: from 50 ohms to Z_{01} , from Z_{01} to Z_{02} , and from Z_{02} back to 50 ohms. The final transition to 50 ohms is required to determine the diode reflection coefficient referred to 50 ohms; there is no length of 50-ohm line at that point.

Although a wave description is needed for a characteristic-impedance change, the voltage-and-current description is much more obvious: Voltage and current are continuous through the transition. In a line of real characteristic impedance Z , the complex amplitudes of the voltage and current are related to the wave complex amplitudes by

$$V = (a + b) \sqrt{Z} \quad \text{and} \quad I = \frac{a - b}{\sqrt{Z}} \quad (10)$$

For these to be continuous through the transition of Figure 8 requires

$$V = (a + b) \sqrt{Z_1} = (\alpha + \beta) \sqrt{Z_2}$$

and

$$I = \frac{a - b}{\sqrt{Z_1}} = \frac{\alpha - \beta}{\sqrt{Z_2}}$$

which have the solution

$$\alpha = 1/2 \left(\sqrt{\frac{Z_1}{Z_2}} + \sqrt{\frac{Z_2}{Z_1}} \right) a + 1/2 \left(\sqrt{\frac{Z_1}{Z_2}} - \sqrt{\frac{Z_2}{Z_1}} \right) b$$

$$\beta = 1/2 \left(\sqrt{\frac{Z_1}{Z_2}} - \sqrt{\frac{Z_2}{Z_1}} \right) a + 1/2 \left(\sqrt{\frac{Z_1}{Z_2}} + \sqrt{\frac{Z_2}{Z_1}} \right) b$$

and T-matrix

$$[T] = \begin{bmatrix} 1/2 \left(\sqrt{\frac{z_1}{z_2}} + \sqrt{\frac{z_2}{z_1}} \right) & 1/2 \left(\sqrt{\frac{z_1}{z_2}} - \sqrt{\frac{z_2}{z_1}} \right) \\ 1/2 \left(\sqrt{\frac{z_1}{z_2}} - \sqrt{\frac{z_2}{z_1}} \right) & 1/2 \left(\sqrt{\frac{z_1}{z_2}} + \sqrt{\frac{z_2}{z_1}} \right) \end{bmatrix}$$

The modified T-matrix used by the program for characteristic-impedance changes is

$$[t] = \frac{2\sqrt{z_1 z_2}}{z_1 + z_2} [T] = \begin{bmatrix} 1 & \frac{z_1 - z_2}{z_1 + z_2} \\ \frac{z_1 - z_2}{z_1 + z_2} & 1 \end{bmatrix} \quad (11)$$

The T-matrix for a length of line having the one-way delay τ emerges easily from the definition of equation (4) and Figure 6:

$$[T] = \begin{bmatrix} e^{-j 2\pi f\tau} & 0 \\ 0 & e^{j2\pi f\tau} \end{bmatrix}$$

The corresponding t-matrix used by the program is

$$[t] = e^{-j2\pi f\tau} [T] = \begin{bmatrix} e^{-j4\pi f\tau} & 0 \\ 0 & 1 \end{bmatrix} \quad (12)$$

t-matrices of the last two forms are used to describe the connecting lines.

Unfortunately the slotted line method of measurement does not provide a means of distinguishing between a reflection coefficient and its conjugate reciprocal: The VSWR, S , on the line is related to the reflection coefficient r at the input reference plane by

$$S = \left| \frac{1 + |r|}{1 - |r|} \right|$$

Note that no change is produced when $|r|$ is replaced by $|r|^{-1}$, and loads having the reflection coefficients r and $1/r^*$ have the same standing wave minimum positions. The program deals with this shortcoming by associating VSWR's greater than unity with passive loads having reflection coefficient magnitudes less than unity, according to

$$|r| = \frac{S - 1}{S + 1}$$

At the voltage minimum point located by the probe, the reflection coefficient is real and negative because the incident and reflected waves are out of phase. Assuming a reflection coefficient r at the input reference plane, as in Figure 9, the reflection coefficient at the probe is

$$r e^{-j4\pi \frac{x}{\lambda_g}} = - \frac{S - 1}{S + 1}$$

from which the expression used by the program is

$$r = - \frac{S - 1}{S + 1} e^{j4\pi \frac{x}{\lambda_g}} \quad (13)$$

Although VSWR's less than unity are never measured, they can certainly be entered into the program, with the following effects:

(1) Changing the sign of S replaces r by $1/r^*$, allowing reflection coefficients greater than one in magnitude to be dealt with.

(2) Replacing S by $1/S$ "changes the sign of $|r|$ ". The result is to interchange standing wave maxima and minima on the slotted line.

According to Figure 10, the reflection coefficient produced by a calibrating load at the output reference plane is

$$\rho = \gamma e^{-j4\pi f\tau} \quad (14)$$

Since γ is real, the load must be pure resistive -- including, perhaps, an open or a short -- at the load plane. But a commercial match and a short at the end of some additional length of line are the most easily acquired accurately-known loads. For the match, $\gamma=0$, and any value may be chosen for τ ; for the short, $\gamma=-1$ at the load plane, and τ is the one-way delay of the additional length of line, if any.

Equations (9) suggest that the output reflection coefficients -- the ρ 's -- and the input reflection coefficients -- the r 's -- should both be as distinct as possible for accurate T-parameter determination. From equation (5), if two calibrating loads produce output reflection coefficients ρ_1 and ρ_2 and the corresponding measured input reflection coefficients are r_1 and r_2 ,

$$r_2 - r_1 = \frac{(T_{11} T_{22} - T_{12} T_{21})(\rho_2 - \rho_1)}{(T_{22} - T_{12} \rho_1)(T_{22} - T_{12} \rho_2)} = \frac{\rho_2 - \rho_1}{(T_{22} - T_{12} \rho_1)(T_{22} - T_{12} \rho_2)}$$

since the T-matrix has a unit determinant. So the input reflection coefficients are distinct if the output reflection coefficients are. An investigation of optimum calibrating loads has not been made, but the evidence at hand suggests either of the following sets of loads:

(1) Three shorts with additional line electrical lengths of 0° , 60° , and 120° .

(2) A match and two shorts with additional line lengths of 0° and 90° .

The simplified, single-load calibration procedure remains to be discussed. The corresponding transducer model, showing incident and reflected wave complex amplitudes, appears in Figure 11. From the figure,

$$\begin{bmatrix} \alpha \\ \beta \end{bmatrix} = \begin{bmatrix} Ke^{-j\theta} & 0 \\ 0 & \frac{1}{K} e^{j\theta} \end{bmatrix} \times \begin{bmatrix} a \\ b \end{bmatrix}$$

If the single calibration load produces the output reflection coefficient $\rho = \beta/\alpha$, the input reflection coefficient that would be measured on the slotted line is

$$r = \frac{b}{a} = \rho K^2 e^{-j2\theta}$$

so that

$$T_{11} = 1/T_{22} = Ke^{-j\theta} = \sqrt{\frac{r}{\rho}}$$

The measurement and the calibration load values then yield

$$[T] = \begin{bmatrix} Ke^{-j\theta} & 0 \\ 0 & \frac{1}{K} e^{j\theta} \end{bmatrix} = \begin{bmatrix} \sqrt{\frac{r}{\rho}} & 0 \\ 0 & \sqrt{\frac{\rho}{r}} \end{bmatrix}$$

The modified T-matrix

$$[t] = \sqrt{\frac{r}{\rho}} [T] = \begin{bmatrix} \frac{r}{\rho} & 0 \\ 0 & 1 \end{bmatrix} \quad (15)$$

is used by the program. Preferably a short, but certainly not a match, should be used as the single calibration load.

2.3 Program Notes


Addition and subtraction of complex numbers are conceptually easier if the numbers are represented in rectangular form; multiplication and division are easier in polar form. But the conversion from rectangular to polar and particularly the conversion back to rectangular consume so much time that these steps should be avoided in a long program: Starting and ending in rectangular form, the 9830A Calculator takes less than 50 milliseconds for a complex multiply without conversion to polar form and back, and it takes more than 500 milliseconds with conversion. For this reason, the present program converts input data to rectangular form, performs all calculations in rectangular form, and only converts to polar at the end if required by the selected printout; the plotter uses the rectangular form as is.

The subroutine at the end of the program, starting with program step 2540 is a rectangular-to-polar conversion used for printout. With one exception, it produces phases which lie in the range $-180^\circ < \text{phase} \leq 180^\circ$ after rounding to one decimal.

The method used by the program to determine which data blocks to print, associated with the second option of Section 1.2, also deserves mention. Five bits are required to specify whether or not to print each of five data blocks, and these five bits are all carried by the single variable P in a way that makes it easy to enter or extract only a specified bit. The first five primes -- 2, 3, 5, 7, 11 -- are used. The "frequency" block, for example, will be printed if and only if P is divisible by 5. At the start, P is reset equal to 0.01 for no printout, 1 for changes only, and $4620=2^2 \cdot 3 \cdot 5 \cdot 7 \cdot 11$ for full printout. If "frequency" block data is entered, for example, P is multiplied by 5. Unless P was initially set equal to 0.01 for no printout, this multiplication leaves P divisible by 5. Mode printout accompanies the point plots if and only if $P > 2310=2 \cdot 3 \cdot 5 \cdot 7 \cdot 11$. Thus starting with $P=1$ so that changes only are printed, mode printout will not occur even if changes are made in all five data blocks. But when full printout is specified and P starts with 4620, mode printout will accompany the point plots even if no data blocks are changed.

3. Acknowledgements

The need for this program was recognized by Sandy Weinreb, who commissioned me to assemble it and will be using it himself in his millimeter-wave front-end work. It would be my pleasure to help others use the program or to modify it to suit their own requirements.



$$G(I) = \frac{I}{V_1}$$

$$C(V) = \begin{cases} c \left(1 - \frac{V}{V_2} \right)^{-N} & , \frac{V}{V_2} < 1/2 \\ c \left[2 \cdot 2^N - \left(\frac{V}{V_2} \right)^{-N} \right] & , \frac{V}{V_2} \geq 1/2 \end{cases}$$

Figure 1. Diode Model




Figure 2. Measurement Block Diagram

NAME	TITLE Figure 3. The Point Plots	OWG. NO.
SMITH CHART FORM 82-BSPR(9-65)	KAY ELECTRIC COMPANY, PINE BROOK, N.J., © 1966. PRINTED IN U.S.A.	DATE

IMPEDANCE OR ADMITTANCE COORDINATES




Figure 4. Printout

```

INPUT OF ROW NO. 8
4 POINTS          FREQUENCY = 25          LAMBDA G = 14.2

CALIBRATIONS:      GAMMA          DELAY          VSWR          X
1) 0.0000          0.0000000     2.150         4.700
2) -1.0000         0.0000000     7.500         8.900
3) -1.0000         0.011364      14.500        4.500

CONNECTING LINES: Z0          DELAY
1) 80.00           0.018779
2) 50.00           0.023760

DIODE MODEL:      R = 18          L = 0.2          C = 5.000000E-06
                  V1 = 28          V2 = 914         N = 0.512

MEASUREMENTS:     I          V          VSWR          X
0.000             0          11.000        11.470
0.100             837        7.000         11.420
0.500             890        3.750         11.680
8.000             1076       3.900         12.220

DIODE MODEL OF ROW NO. 8
DIODE D-C          0          DIODE RF
I          V          ABS(GAMMA)          ARG(GAMMA)
0.000     0          0.9988             -4.6
0.100     837        0.6937             -5.3
0.500     890        0.2831             36.4
8.000     1076       0.5430             108.5

MEASUREMENTS OF ROW NO. 8
SLOTTED LINE      X          DIODE RF
VSWR          X          ABS(GAMMA)          ARG(GAMMA)
11.000        11.470          0.9126             -0.5
7.000         11.420          0.7017             -8.5
3.750         11.680          0.3058             28.1
3.900         12.220          0.4854             99.0

```

Figure 5. Options and Remarks

```
30 REM: UNITS ARE MILLIAMPS, MILLIVOLTS, OHMS, GHZ, NANOHENRIES, NANOFARADS.
40 DEG
```

```
80 DISP "DATA ROW NO.=";
90 INPUT L
```

```
100 DISP "PRINTOUT? P=";
110 REM: P=0 -- NO PRINTOUT      P=1 -- CHANGES ONLY      P=2 -- PRINT ALL
120 INPUT P
```

```
180 DISP "CONNECTING LINE CHANGE? K=";
190 REM: K=0 -- EXIT CHANGES    K=1 -- CHANGE ELSEWHERE    K=2 -- CHANGE HERE
200 INPUT K
```

```
280 DISP "DIODE MODEL CHANGE? K=";
290 INPUT K
```

```
390 DISP "FREQUENCY CHANGE? K=";
400 INPUT K
```

```
460 DISP "CALIBRATION CHANGE? M=";
470 REM: M=0 -- EXIT CHANGES    M=1 -- CHANGE ELSEWHERE
480 REM: M=2 -- CHANGE MEASUREMENTS M=3 -- CHANGE ALL
490 INPUT M
```

```
620 DISP "MEASUREMENT CHANGE? M=";
630 REM: M=0 -- EXIT CHANGES    M=1 -- CHANGE D-C
640 REM: M=2 -- CHANGE RF        M=3 -- CHANGE BOTH
650 INPUT M
```

```
1130 DISP "OUTPUT? M=";
1140 REM: M=0 -- NO OUTPUT        M=1 -- DIODE MODEL
1150 REM: M=2 -- MEASUREMENTS    M=3 -- BOTH
1160 INPUT M
```





Figure 6. Waves Associated with the Transducer




Figure 7. Transducers in Tandem




Figure 8. A Characteristic Impedance Transition




Figure 9. Waves on the Slotted Line




Figure 10. Waves at a Calibrating Load




Figure 11. Simplified Transducer Model

PROGRAM

```

10 COM DC[6],FC[8],LC[8],JC[8],RC[5],TC[5],SC[8,18],XC[8,18],IC[8,15],VC[8,15]
20 DIM AC[11],BC[11]
30 REM: UNITS ARE MILLIAMPS, MILLIVOLTS, OHMS, GHZ, NANOHENRIES, NANOFARADS.
40 DEG
50 K=1.14927
60 SCALE -K,K,-K,K
70 LABEL (*,0.6,1,0)
80 DISP "DATA ROW NO.=";
90 INPUT L
100 DISP "PRINTOUT? P=";
110 REM: P=0 -- NO PRINTOUT      P=1 -- CHANGES ONLY      P=2 -- PRINT ALL
120 INPUT P
130 IF P=0 THEN 170
140 IF P=1 THEN 180
150 P=4620
160 GOTO 180
170 P=0.01
180 DISP "CONNECTING LINE CHANGE? K=";
190 REM: K=0 -- EXIT CHANGES    K=1 -- CHANGE ELSEWHERE    K=2 -- CHANGE HERE
200 INPUT K
210 IF K=0 THEN 820
220 IF K=1 THEN 280
230 P=2*P
240 FOR I=1 TO 2
250 DISP "CONNECTING LINE NO."I"Z0,T=";
260 INPUT RC[I+3],TC[I+3]
270 NEXT I
280 DISP "DIODE MODEL CHANGE? K=";
290 INPUT K
300 IF K=0 THEN 820
310 IF K=1 THEN 390
320 P=3*P
330 DISP "R,L=";
340 INPUT DC[1],DC[2]
350 DISP "C,N=";
360 INPUT DC[3],DC[4]
370 DISP "V1,V2=";
380 INPUT DC[5],DC[6]
390 DISP "FREQUENCY CHANGE? K=";
400 INPUT K
410 IF K=0 THEN 820
420 IF K=1 THEN 460
430 P=5*P
440 DISP "FREQUENCY, LAMBDA G=";
450 INPUT FC[L],LC[L]
460 DISP "CALIBRATION CHANGE? M=";
470 REM: M=0 -- EXIT CHANGES      M=1 -- CHANGE ELSEWHERE
480 REM: M=2 -- CHANGE MEASUREMENTS M=3 -- CHANGE ALL
490 INPUT M
500 IF M=0 THEN 820

```

```

510 FOR K=1 TO 3
520 GOTO M OF 620,530,560
530 DISP "LOAD NO."K"VSWR,X=";
540 INPUT S[L,K],X[L,K]
550 GOTO 580
560 DISP "LOAD NO."K"GAMMA,T,VSWR,X=";
570 INPUT R[K],T[K],S[L,K],X[L,K]
580 REM: SET NO.2 VSWR=999 FOR SINGLE-LOAD CALIBRATION.
590 IF K=2 AND S[L,2]=999 THEN 610
600 NEXT K
610 P=7*P
620 DISP "MEASUREMENT CHANGE? M=";
630 REM: M=0 -- EXIT CHANGES      M=1 -- CHANGE D-C
640 REM M=2 -- CHANGE RF          M=3 -- CHANGE BOTH
650 INPUT M
660 IF M=0 THEN 820
670 FOR K=1 TO 15
680 GOTO M OF 690,720,750
690 DISP "I,V=";
700 INPUT I[L,K],V[L,K]
710 GOTO 770
720 DISP "VSWR,X=";
730 INPUT S[L,K+3],X[L,K+3]
740 GOTO 770
750 DISP "I,V,VSWR,X=";
760 INPUT I[L,K],V[L,K],S[L,K+3],X[L,K+3]
770 REM: SET I<0 OR VSWR=999 TO EXIT LOOP.
780 IF I[L,K]<0 OR S[L,K+3]=999 THEN 800
790 NEXT K
800 J[L]=K-1
810 P=11*P
820 IF P/5>INT(P/5) THEN 860
830 PRINT "INPUT OF ROW NO."L
840 PRINT J[L]"POINTS"TAB26"FREQUENCY ="F[L];TAB54"LAMBDA G ="L[L]
850 PRINT
860 IF P/7>INT(P/7) THEN 940
870 PRINT "CALIBRATIONS:"TAB23"GAMMA"TAB38"DELAY"TAB53"VSWR"TAB69"X"
880 FOR I=1 TO 3
890 FORMAT F7.0,")",F21.4,F15.6,2F14.3
900 WRITE (15,890)I,R[I],T[I],S[L,I],X[L,I]
910 IF S[L,2]=999 THEN 930
920 NEXT I
930 PRINT
940 IF P/2>INT(P/2) THEN 1010
950 PRINT "CONNECTING LINES:"TAB25"Z0"TAB38"DELAY"
960 FOR I=1 TO 2
970 FORMAT F7.0,")",F21.2,F15.6
980 WRITE (15,970)I,R[I+3],T[I+3]
990 NEXT I
1000 PRINT
1010 IF P/3>INT(P/3) THEN 1050
1020 PRINT "DIODE MODEL:"TAB19"R ="D[1];TAB37"L ="D[2];TAB55"C ="D[3]
1030 PRINT TAB18"V1 ="D[5];TAB36"V2 ="D[6];TAB55"N ="D[4]
1040 PRINT

```

```

1050 IF P/11>INT(P/11) THEN 1130
1060 PRINT "MEASUREMENTS:"TAB25"I"TAB39"V"TAB53"VSWR"TAB69"X"
1070 FOR I=1 TO J[L]
1080 FORMAT F28.3,F13.0,F17.3,F14.3
1090 WRITE (15,1080)I[L,I],V[L,I],S[L,I+3],X[L,I+3]
1100 NEXT I
1110 PRINT
1120 PRINT
1130 DISP "OUTPUT? M=";
1140 REM: M=0 -- NO OUTPUT      M=1 -- DIODE MODEL
1150 REM M=2 -- MEASUREMENTS  M=3 -- BOTH
1160 INPUT M
1170 GOTO M+1 OF 2530,1180,1490,1180
1180 IF P <= 2310 THEN 1220
1190 PRINT "DIODE MODEL OF ROW NO."L
1200 PRINT TAB8"DIODE D-C"TAB33"O"TAB51"DIODE RF"
1210 PRINT TAB4"I"TAB21"V"TAB41"ABS(GAMMA)"TAB58"ARG(GAMMA)"
1220 FOR I=1 TO J[L]
1230 G=I[L,I]/D[5]
1240 W=V[L,I]/D[6]
1250 IF W<0.5 THEN 1280
1260 B=2*PI*F[L]*D[3]*(2*2↑D[4]-W↑(-D[4]))
1270 GOTO 1290
1280 B=2*PI*F[L]*D[3]/(1-W)↑D[4]
1290 D=G*G+B*B
1300 R=(G/D+D[1])/50
1310 X=(-B/D+2*PI*F[L]*D[2])/50
1320 D=(R+1)*(R+1)+X*X
1330 A=(R*R+X*X-1)/D
1340 B=2*X/D
1350 IF P <= 2310 THEN 1390
1360 GOSUB 2540
1370 FORMAT F7.3,F16.0,F26.4,F17.1
1380 WRITE (15,1370)I[L,I],V[L,I],R,G
1390 PLOT A,B
1400 CPLOT -0.3,-0.3
1410 LABEL (*)"0"
1420 IPLOT 0,0
1430 NEXT I
1440 IF P <= 2310 THEN 1470
1450 PRINT
1460 PRINT
1470 PEN
1480 IF M<2 THEN 2530
1490 IF P <= 2310 THEN 1530
1500 PRINT "MEASUREMENTS OF ROW NO."L
1510 PRINT TAB7"SLOTTED LINE"TAB33"X"TAB51"DIODE RF"
1520 PRINT "  VSWR"TAB21"X"TAB41"ABS(GAMMA)"TAB58"ARG(GAMMA)"
1530 IF S[L,2]#999 THEN 1610
1540 G=720*(F[L]*T[1]+X[L,1])/L[L]
1550 R=(1-S[L,1])/(1+S[L,1])/R[1]
1560 A[5]=R*COSG
1570 B[5]=R*SING
1580 A[3]=B[3]=B[4]=A[6]=B[6]=0

```

```

1590 A[4]=1
1600 GOTO 2030
1610 FOR I=1 TO 3
1620 G=720*F[L]*T[I]
1630 K=I+1
1640 A[K]=R[I]*COSG
1650 B[K]=-R[I]*SING
1660 K=K-1
1670 R=(1-S[L,I])/(1+S[L,I])
1680 G=720*X[L,I]/L[L]
1690 A[K]=R*COSG
1700 B[K]=R*SING
1710 NEXT I
1720 FOR K=2 TO 4 STEP 2
1730 FOR I=1 TO 2
1740 A[I+K+5]=A[I+K]-A[I]
1750 B[I+K+5]=B[I+K]-B[I]
1760 NEXT I
1770 NEXT K
1780 FOR I=3 TO 1 STEP -1
1790 M=I+1
1800 A[I+4]=A[M-1]*A[M]-B[M-1]*B[M]
1810 B[I+4]=A[M-1]*B[M]+B[M-1]*A[M]
1820 NEXT I
1830 FOR I=6 TO 7
1840 A[I]=A[I]-A[5]
1850 B[I]=B[I]-B[5]
1860 NEXT I
1870 FOR I=4 TO 5
1880 A[I]=A[7]*A[13-I]-B[7]*B[13-I]
1890 B[I]=A[7]*B[13-I]+B[7]*A[13-I]
1900 A[I]=A[6]*A[15-I]-B[6]*B[15-I]-A[I]
1910 B[I]=A[6]*B[15-I]+B[6]*A[15-I]-B[I]
1920 NEXT I
1930 A[6]=A[9]*A[10]-B[9]*B[10]
1940 B[6]=A[9]*B[10]+B[9]*A[10]
1950 A[6]=A[8]*A[11]-B[8]*B[11]-A[6]
1960 B[6]=A[8]*B[11]+B[8]*A[11]-B[6]
1970 A[7]=A[2]*A[6]-B[2]*B[6]-A[4]
1980 B[7]=A[2]*B[6]+B[2]*A[6]-B[4]
1990 A[3]=A[1]*A[7]-B[1]*B[7]
2000 B[3]=A[1]*B[7]+B[1]*A[7]
2010 A[3]=A[2]*A[5]-B[2]*B[5]+A[3]
2020 B[3]=A[2]*B[5]+B[2]*A[5]+B[3]
2030 REM: MOD. T-MATRIX TO OUTPUT REF. IN A,B 5,6,3,4.
2040 FOR K=1 TO 3
2050 IF K=1 THEN 2130
2060 G=720*F[L]*T[K+2]
2070 A=COSG
2080 B=-SING
2090 FOR I=1 TO 2
2100 A[I+4]=A*A[I]-B*B[I]
2110 B[I+4]=A*B[I]+B*A[I]
2120 NEXT I

```

```

2130 GOTO K OF 2140,2160,2180
2140 R=(50-R[4])/(50+R[4])
2150 GOTO 2190
2160 R=(R[4]-R[5])/(R[4]+R[5])
2170 GOTO 2190
2180 R=(R[5]-50)/(R[5]+50)
2190 FOR I=1 TO 4
2200 M=I
2210 IF I>2 THEN 2230
2220 M=M+4
2230 A[I]=A[M]+R*A[I+2]
2240 B[I]=B[M]+R*B[I+2]
2250 NEXT I
2260 NEXT K
2270 REM: MOD. T-MATRIX TO DIODE REF. IN A,B 1 THRU 4.
2280 FOR K=4 TO J[L]+3
2290 R=(1-S[L,K])/(1+S[L,K])
2300 G=720*X[L,K]/L[L]
2310 A=R*COSG
2320 B=R*SING
2330 M=6
2340 FOR I=2 TO 4 STEP 2
2350 A[M]=A*A[I]-B*B[I]+A[I-1]
2360 B[M]=A*B[I]+B*A[I]+B[I-1]
2370 M=5
2380 NEXT I
2390 D=A[6]*A[6]+B[6]*B[6]
2400 H=(A[5]*A[6]+B[5]*B[6])/D
2410 B=(B[5]*A[6]-A[5]*B[6])/D
2420 IF P <= 2310 THEN 2460
2430 GOSUB 2540
2440 FORMAT F7.3,F17.3,F25.4,F17.1
2450 WRITE (15,2440)S[L,K],X[L,K],R,G
2460 PLOT A,B,1
2470 CPLOT -0.3,-0.3
2480 LABEL (*)"X"
2490 NEXT K
2500 IF P <= 2310 THEN 2530
2510 PRINT
2520 PRINT
2530 END
2540 R=SQR(A*A+B*B)
2550 G=90*SGNB
2560 IF A=0 THEN 2620
2570 G=ATN(B/A)
2580 IF A>0 THEN 2620
2590 G=G+180
2600 IF G<180.05 THEN 2620
2610 G=G-360
2620 RETURN

```


September 24, 1976

ADDENDUM NUMBER 1 TO ELECTRONICS DIVISION INTERNAL REPORT #171

(This page constitutes an addition to Section 1.3 and should be inserted after page 6.)

In some cases the operator will be certain that both the transducer and either a calibrating load or perhaps even an unknown load are lossless, yet he will have measured a finite VSWR at the input reference plane, indicating losses. These losses may be in the slotted line itself, or they may result from an imperfection in the method of measurement. In either case, it is possible to override the measurement, making the program take the entire load connected to the slotted line to be lossless: The measured probe position is entered as is, but the measured VSWR is replaced by a large number, ideally infinity, for the VSWR entry. Since the 9830A Calculator performs 12-digit arithmetic, any number equal to or greater than 10^{12} produces an input reflection coefficient magnitude equal to unity to all twelve digits, just as an infinite entry would have done if that were possible. There is not room in the fixed-point formats used by the program for the printout of VSWR's as large as 10^{12} . "Calibrations" block and "measurements" block VSWR's $\geq 10^{12}$ are printed in a floating-point format having three digits to the right of the decimal point and using a total of 10 spaces. Since only 7 spaces are allowed for VSWR in the Mode (2) printout, VSWR's $\geq 10^{12}$ here are represented by dollar signs, but the actual values used can be read from the "measurements" block printout.

